

IIC2323 — Construcción de Compiladores

Especificación para un subconjunto de Sather 1.1

Primer semestre 2014

Índice

1. Notación	1
2. Estructura léxica	2
2.1. Identificadores	2
2.2. Keywords	2
2.3. Símbolos especiales	2
2.4. Expresiones literales	2
2.5. Comentarios	2
3. Estructura sintáctica	3

1. Notación

El lenguaje es especificado utilizando gramáticas libres de contexto, donde al lado derecho de las producciones se utilizan expresiones regulares con la siguiente notación:

1. $[\varphi]$ es equivalente a $\varphi + \varepsilon$, es decir, que φ es opcional.
2. $\{\varphi\}$ corresponde a φ^* , es decir que φ se puede repetir una cantidad arbitraria de veces.
3. $\varphi|\psi$ es equivalente a $\varphi + \psi$, es decir que se escoge o bien φ o ψ .

Por ejemplo, la gramática:

$$\begin{aligned} \textit{identifier} &\rightarrow [\$]\textit{letter}\{\textit{letter}|\textit{digit}\} \\ \textit{letter} &\rightarrow a|b|c \\ \textit{digit} &\rightarrow 0|1 \end{aligned}$$

indica que un identificador puede comenzar opcionalmente con \$, luego sigue con una letra, y termina con una cantidad arbitraria de letras y dígitos. Algunos identificadores válidos según esta son: \$a00b10a, b, y \$c0, mientras que 10, y a\$ no lo son.

2. Estructura léxica

2.1. Identificadores

Los identificadores dan nombre a clases, métodos, atributos y variables.

$$\begin{aligned} identifier &\rightarrow letter\{letter|decimal_digit|_|\} \\ uppercase_identifier &\rightarrow uppercase_letter\{uppercase_letter|decimal_digit|_|\} \\ iter_name &\rightarrow identifier! \\ letter &\rightarrow lowercase_letter|uppercase_letter \\ lowercase_letter &\rightarrow \mathbf{a|b|c|d|e|f|g|h|i|j|k|l|m|n|o|p|q|r|s|t|u|v|w|x|y|z} \\ uppercase_letter &\rightarrow \mathbf{A|B|C|D|E|F|G|H|I|J|K|L|M|N|O|P|Q|R|S|T|U|V|W|X|Y|Z} \\ decimal_digit &\rightarrow \mathbf{0|1|2|3|4|5|6|7|8|9} \end{aligned}$$

2.2. Keywords

Los keywords identifican elementos sintácticos fundamentales, y no pueden ser utilizados como identificadores.

$$\begin{aligned} keyword &\rightarrow \mathbf{and|attr|break!|class|else|elsif|end|false|if|include|is|loop|new|or|once} \\ &\quad \mathbf{private|quit|return|SAME|self|then|true|typecase|void|when|while!|yield} \end{aligned}$$

2.3. Símbolos especiales

Otros símbolos utilizados:

$$special_symbol \rightarrow (|) | \{ | \} | , | . | ; | : | + | - | * | / | \% | < | <= | = | /= | > | >= | \sim | := | :: | \#$$

2.4. Expresiones literales

Valores literales para tipos básicos.

$$\begin{aligned} bool_literal_expression &\rightarrow \mathbf{true|false} \\ char_literal_expression &\rightarrow '(ISO_character|\backslash escape_seq)'' \\ escape_seq &\rightarrow a|b|f|n|r|t|v|'|''|octal_digit\{octal_digit\}|\backslash \\ str_literal_expression &\rightarrow "\{(ISO_character|\backslash escape_seq)\}"' "\{(ISO_character|\backslash escape_seq)\}"' \\ int_literal_expression &\rightarrow [-]decimal_digit\{decimal_digit\} \end{aligned}$$

2.5. Comentarios

Los comentarios comienzan con `--` y continúan hasta el siguiente salto de línea.

3. Estructura sintáctica

```
source_file → [class]{; [class]}
```

```
type_specifier → uppercase_identifier | SAME
```

```
routine_argument → identifier_list : type_specifier
```

```
identifier_list → identifier{, identifier}
```

```
subtyping_clause → < type_specifier_list
```

```
type_specifier_list → type_specifier{, type_specifier}
```

```
class → class uppercase_identifier[subtyping_clause] is  
      [class_element]{; [class_element]} end
```

```
class_element → attr_definition|routine_definition|iter_definition|include_clause
```

```
attr_definition → [private] attr identifier_list : type_specifier
```

```
routine_definition → [private]identifier[(routine_argument{, routine_argument})][ : type_specifier]  
                   is statement_list end
```

```
iter_definition → [private]iter_name[([once]routine_argument{, [once]routine_argument})]  
                  [ : type_specifier] is statement_list end
```

```
include_clause → include type_specifier[feature_modifier{, feature_modifier}]
```

```
feature_modifier → identifier - identifier|iter_name - iter_name
```

```
statement_list → [statement]{; [statement]}
```

```
statement → declaration_statement|assign_statement|if_statement|return_statement
```

```
statement → typecase_statement|expression_statement|loop_statement
```

```
statement → yield_statement|quit_statement
```

```
declaration_statement → identifier_list : type_specifier
```

```
assign_statement → (expression|identifier : [type_specifier]) := expression
```

```
if_statement → if expression then statement_list{elsif expression then statement_list}  
              [else statement_list] end
```

```
return_statement → return [expression]
```

```
typecase_statement → typecase identifier when type_specifier then statement_list  
                    {when type_specifier then statement_list}[else statement_list] end
```

```
expression_statement → expression
```

```
expression → bool_literal_expression|char_literal_expression
```

```
expression → str_literal_expression|int_literal_expression
```

```
expression → self_expression|local_expression|call_expression|void_expression|or_expression
```

```
expression → void_test_expression|new_expression|create_expression|and_expression
```

```
expression → sugar_expression|while!_expression|break!_expression
```

```
self_expression → self
```

```
local_expression → identifier
```

```
call_expression → expression.(identifier|iter_name)[(expression{, expression})]
```

```
void_expression → void
```

```
void_test_expression → void(expression)
```

$new_expression \rightarrow \mathbf{new}$
 $create_expression \rightarrow \#[type_specifier][(expression\{, expression\})]$
 $and_expression \rightarrow expression \mathbf{and} expression$
 $or_expression \rightarrow expression \mathbf{or} expression$
 $sugar_expression \rightarrow expression \mathit{binary_op} expression | -expression$
 $sugar_expression \rightarrow (expression) | \sim expression$
 $binary_op \rightarrow + | - | * | / | \% | < | <= | = | / = | > | >=$
 $loop_statement \rightarrow \mathbf{loop} statement_list \mathbf{end}$
 $yield_statement \rightarrow \mathbf{yield} [expression]$
 $quit_statement \rightarrow \mathbf{quit}$
 $while!_expression \rightarrow \mathbf{while!}(expression)$
 $break!_expression \rightarrow \mathbf{break!}$

Referencias

- [1] Stoutamire, David, and Stephen Omohundro. “The sather 1.1 specification.” INTERNATIONAL COMPUTER SCIENCE INSTITUTE-PUBLICATIONS-TR (1996).